

Лекция 7.

Работа с формами (часть 2)

РХТУ им. Д.И. Менделеева

Каф. ИКТ

Курс создал: ст. преп. А.М. Васецкий

2013 г

Переключатель

Элемент управления **OptionButton (переключатель)** позволяет выбрать один из нескольких взаимоисключающих параметров или действий. Переключатели обычно отображаются группами, обеспечивая возможность выбора альтернативного варианта. Группы формируются внутри рамки (**Frame**).

Элемент управления **Frame (рамка)** создается с помощью кнопки **Рамка (Frame)**. Он используется для визуальной группировки элементов управления. Основным свойством рамки является **Capture**, отображающее надпись рамки.

Value	Возвращает True , если переключатель выбран и False в противном случае
Enabled	Допустимые значения: True (пользователь может выбрать переключатель) и False , (в противном случае)
Visible	Допустимые значения: True (переключатель отображается во время выполнения программы) и False (в противном случае)
Capture	Надпись, отображаемая рядом с переключателем

Полоса прокрутки

Value	Возвращает текущее значение полосы прокрутки (только целые неотрицательные числа)
Min	Минимальное значение полосы прокрутки (только целые неотрицательные числа)
Max	Максимальное значение полосы прокрутки (только целые неотрицательные числа)
SmallChange	Устанавливает шаг изменения значения при щелчке по одной из стрелок полосы прокрутки
Enabled	Допустимые значения: True (пользователь может изменить значение полосы прокрутки) и False (в противном случае)
Visible	Допустимые значения: True (полоса прокрутки отображается во время выполнения программы) и False ; (в противном случае)

Пример

```
Private Sub ScrollBar1_Change()  
'Прямая синхронизация с текстовым полем  
TextBox1.Value = ScrollBar1.Value / 10  
End Sub
```

```
Private Sub TextBox1_Exit(ByVal Cancel As MSForms.ReturnBoolean)  
'Обратная синхронизация текстового поля с полосой прокрутки  
Call Check  
With ScrollBar1  
If dVar1 * 10 <= .Max And dVar1 * 10 >= .Min Then  
.Value = dVar1 * 10  
Else  
.Value = (.Max + .Min) / 2  
MsgBox "Некорректное число!", vbCritical  
End If  
End With  
End Sub
```

```
Private Sub UserForm_Initialize()  
TextBox1.Value = 5  
With ScrollBar1  
.Min = 0  
.Max = 100  
.Value = 50  
End With  
End Sub
```

```
Sub Check() 'проверка textbox  
Dim Str As String 'промежуточная переменная  
Str = Replace(TextBox1.Value, ".", ",")  
If IsNumeric(Str) Then  
dVar1 = CDb1(Str)  
Else  
MsgBox "Некорректное значение", vbCritical  
End If  
End Sub
```

```
Option Explicit
```

```
Private Sub CommandButton1_Click()  
Call Check  
hide  
End Sub
```

```
Private Sub CommandButton2_Click()  
hide  
End Sub
```

```
Option Explicit
```

```
Public dVar1 As Double 'Целевая переменная значение
```

```
Sub ScrollBarTest()  
UserForm7.Show  
MsgBox "Считано число" & vbCrLf & _  
dVar1, vbInformation  
End Sub
```

Прокрутка

Установите параметр



OK

Отмена

Счётчик (SpinButton)

Счётчик работает аналогично полосе прокрутки (только без ползунка)

Properties - SpinButton1	
SpinButton1 SpinButton	
Alphabetic Categorized	
(Name)	SpinButton1
BackColor	<input type="checkbox"/> &H8000000F&
Delay	50
Enabled	True
ForeColor	<input checked="" type="checkbox"/> &H80000012&
Height	102
Left	535,75
Max	100
Min	0
MouseIcon	(None)
MousePointer	0 - fmMousePointerDefault
Orientation	-1 - fmOrientationAuto
SmallChange	1
Top	139,625
Value	0
Visible	True
Width	96,375

Option Explicit

```
Private Sub SpinButton1_Change()  
    TextBox1.Value = SpinButton1.Value  
End Sub
```

Флажок (CheckBox)

Этот элемент управления предоставляет пользователю возможность выбора. Флажок обычно имеет два состояния: **установленное** и **сброшенное**, но может настраиваться на выбор из трех альтернатив. Флажок имеет те же свойства **value**, **Enabled**, **visible** и **Capture**, что и переключатель. Кроме того, флажок обладает уникальным свойством **Triplestate**, позволяющим производить выбор из трех альтернатив. Свойство **Triplestate** может принимать два значения: **False** (выбор из двух альтернатив (**True** и **False**)) и **True** (выбор из трех альтернатив **True**, **False** и **Null**).

Выключатель (ToggleButton)

Создается с помощью кнопки **Выключатель (ToggleButton)**.
Предоставляет те же возможности, что и флажок.

```
Option Explicit
```

```
Private Sub CommandButton1_Click()  
Hide  
End Sub
```

```
Private Sub CheckBox1_Click()  
ToggleButton1.Value = CheckBox1.Value  
End Sub
```

```
Private Sub CheckBox2_Click()  
ToggleButton2.Value = CheckBox2.Value  
End Sub
```

```
Private Sub CheckBox3_Click()  
' Эта функция не срабатывает если после щелчка  
' устанавливается неопределённое положение флажка  
ToggleButton3.Value = CheckBox3.Value  
End Sub
```

```
Private Sub ToggleButton1_Click()  
CheckBox1.Value = ToggleButton1.Value  
End Sub
```


```
Private Sub ToggleButton2_Click()  
CheckBox2.Value = ToggleButton2.Value  
End Sub
```

```
Private Sub ToggleButton3_Click()  
' Эта функция не срабатывает если после щелчка  
' устанавливается неопределённое положение выключателя  
CheckBox3.Value = ToggleButton3.Value  
End Sub
```


```
Private Sub UserForm_Initialize()
```

```
 ' Выключатели  
ToggleButton1.Value = True  
ToggleButton2.Value = False  
With ToggleButton3  
 .TriState = True 'разрешено неопределённое состояние  
 .Value = Null 'Задаём неопределённое состояние  
End With
```

```
 ' Флажки  
CheckBox1.Value = True  
CheckBox2.Value = False  
With CheckBox3  
 .TriState = True 'разрешено неопределённое состояние  
 .Value = Null 'Задаём неопределённое состояние  
End With  
End Sub
```



Пример



Пример

```
Const DefaultFontSize As Byte = 12
Const MinSize = 6, MaxSize = 99
```

```
Private Sub CheckBox1_Click()
```

```
Frame1.Enabled = Not CheckBox1.Value 'закрывает возможность выбора
OptionButton1.Enabled = Frame1.Enabled
OptionButton2.Enabled = Frame1.Enabled
OptionButton3.Enabled = Frame1.Enabled
OptionButton3.Value = True 'устанавливается цвет по умолчанию в любом случае
End Sub
```

```
Private Sub CheckBox2_Click()
```

```
Label1.Font.Bold = CheckBox2.Value
End Sub
```

```
Private Sub CheckBox3_Click()
```

```
Label1.Font.Italic = CheckBox3.Value
End Sub
```

```
Private Sub CheckBox4_Click()
```

```
Label1.Font.Underline = CheckBox4.Value
End Sub
```

```
Private Sub CommandButton2_Click()
```

```
hide
End Sub
```

```
Private Sub OptionButton1_Click()
```

```
If OptionButton1.Value Then Label1.ForeColor = vbRed 'шрифт красный
End Sub
```

```
Private Sub OptionButton2_Click()
```

```
If OptionButton2.Value Then Label1.ForeColor = vbBlue 'шрифт синий
End Sub
```

```
Option Explicit
```

```
Sub FontSet()
```

```
'Установки для шрифта
```

```
UserForm6.Show
```

```
End Sub
```

Код
ОСНОВНОГО
МОДУЛЯ

Пример

```
Private Sub OptionButton3_Click()  
If OptionButton3.Value Then Label1.ForeColor = vbBlack 'шрифт чёрный  
End Sub
```

```
Private Sub SpinButton1_Change()  
With SpinButton1  
If .Value < MinSize Then .Value = MinSize 'проверка размера шрифта  
If .Value > MaxSize Then .Value = MaxSize 'проверка размера шрифта  
TextBox1.Value = .Value  
Label1.Font.Size = .Value 'форматируем образец текста нужным размером шрифта  
End With  
End Sub
```

```
Private Sub TextBox1_AfterUpdate()  
With TextBox1  
.Value = Replace(.Value, ".", ",") 'замена десятичного разделителя  
If IsNumeric(.Value) Then  
If .Value < MinSize Then .Value = MinSize 'проверка размера шрифта  
If .Value > MaxSize Then .Value = MaxSize 'проверка размера шрифта  
Else  
MsgBox "Нечисловое значение!", vbCritical  
.Value = DefaultFontSize  
End If  
SpinButton1.Value = .Value  
Label1.Font.Size = .Value 'форматируем образец текста нужн  
End With  
End Sub
```

```
Private Sub UserForm_Initialize()  
'установки по умолчанию  
TextBox1.Value = DefaultFontSize  
SpinButton1.Value = DefaultFontSize  
OptionButton3.Value = True 'установки по умолчанию  
CheckBox1.Value = False  
CheckBox2.Value = False  
CheckBox3.Value = False  
CheckBox4.Value = False  
With Label1  
.Font.Size = SpinButton1.Value  
.ForeColor = vbBlack  
End With  
OptionButton1.ForeColor = vbRed  
OptionButton2.ForeColor = vbBlue  
OptionButton3.ForeColor = vbBlack  
End Sub
```

Редактирование ссылок (RefEdit)

Обычно используется для ввода ссылок на ячейки или диапазоны. Основным свойством является **Text**, возвращающее строку, введенную в поле ввода этого элемента управления.


Option Explicit

```
Sub RefEditShow()  
Dim Str As String  
  
With UserForm1  
    .Show  
    Str = .RefEdit1.Value  
    ' Str = .RefEdit1.Text  
End With  
MsgBox "Считано: " & vbCrLf & Str, vbInformation  
End Sub
```




Option Explicit

```
Private Sub CommandButton1_Click()  
Hide  
End Sub
```



Набор страниц (MultiPage)

Этот элемент управления реализует многостраничные диалоговые окна. Заголовки страниц отображаются на вкладках. Переход от страницы к странице осуществляется выбором вкладки посредством щелчка кнопкой мыши.



Свойства

Value и
BoundValue

Возвращают номер активной страницы.

Нумерация производится с нуля

MultiRow

Допустимые значения: **True** (если ярлыки не помещаются в одну строку, то они выводятся в несколько строк) и **False** (если ярлыки не помещаются в одну строку, то появляется полоса прокрутки, позволяющая переходить от страницы к странице)

SelectedItem

Возвращает выбранную страницу

Методы семейства Pages

Семейство **Pages**, содержащее все страницы, входящие в объект **MultiPage**, имеет единственное свойство **count**, возвращающее число элементов семейства. Кроме того, у семейства **Pages** имеются следующие методы:

Add	<p>Создает новую страницу. Синтаксис: Set Object = object.Add([Name[, Caption[, index]])</p> <ul style="list-style-type: none">• Object – семейство Pages• Name – имя страницы• Caption – текст, отображаемый на ярлыке страницы• index – номер страницы, нумерация страниц производится с 0
Clear	Удаляет все страницы из семейства Pages
Remove	Удаляет страницу из семейства Pages
Item	<p>Возвращает страницу со специфицированным индексом. Синтаксис: Set Object = object.Item(collectionindex)</p>

Пример

```
Option Explicit
```

```
Private Sub CheckBox1_Click()  
If CheckBox1.Value Then  
    MultiPage1.Style = fmTabStyleButtons 'в виде кнопок  
Else  
    MultiPage1.Style = fmTabStyleTabs 'в виде вкладок  
End If  
End Sub
```

```
Private Sub CheckBox2_Click()  
If CheckBox2.Value Then  
    'убираем графу напрямую  
    MultiPage1.Page3.Visible = False  
Else  
    'показываем графу, используя её порядковый номер  
    MultiPage1.Pages(2).Visible = True  
End If  
End Sub
```

```
Private Sub CommandButton1_Click()  
Hide  
End Sub
```


```
Private Sub OptionButton1_Click()  
MultiPage1.TabOrientation = fmTabOrientationTop  
End Sub
```

```
Private Sub OptionButton2_Click()  
MultiPage1.TabOrientation = fmTabOrientationBottom  
End Sub
```

```
Private Sub OptionButton3_Click()  
MultiPage1.TabOrientation = fmTabOrientationRight  
End Sub
```

```
Private Sub OptionButton4_Click()  
MultiPage1.TabOrientation = fmTabOrientationLeft  
End Sub
```

```
Private Sub UserForm_Initialize()  
    'в данном случае With нужен в основном для отображения  
    'при вводе списка инструментов, содержащихся на форме  
    With UserForm12  
        .CheckBox1 = False 'в виде вкладок  
        .CheckBox2 = False 'показывать графу  
        .OptionButton1 = True 'вкладки сверху  
    End With  
End Sub
```



Набор вкладок (TabStrip)

Он позволяет создать несколько вкладок в диалоговом окне. Объект **TabStrip** содержит в себе семейство **Tabs**, представляющее собой набор всех вкладок. Объект **Tabstrip** и семейство **Tabs** обладают теми же свойствами и методами, что и объект **MuitiPage** и семейство **Pages**.

TabStrip НЕ ЯВЛЯЕТСЯ КОНТЕЙНЕРОМ, если добавить его к пользовательской форме и затем поместить во внутреннюю область, например, кнопку, эта кнопка не будет перемещаться вместе с **TabStrip**. Следовательно, самого по себе **TabStrip** недостаточно, чтобы обеспечить вывод нужной страницы при щелчке по закладке. Для этого необходим еще один элемент управления, являющийся контейнером.

Лучше всего создать по отдельному контейнеру для каждого объекта **Tab** и написать код, который будет выводить и убирать эти контейнеры в зависимости от того, по какой закладке щелкнул пользователь. По размеру они должны соответствовать внутренней области **TabStrip**.

Пример

```
Option Explicit
Type FIO
    Surname As String
    Name As String
    Patronymic As String
End Type
```

```
Sub TabStripShow()
' Демонстрирует вкладки
Dim Ya As FIO
Dim i As Byte
10:
UserForm13.Show
With UserForm13
    Ya.Name = .TextBox2.Text
    Ya.Patronymic = .TextBox3.Text
    Ya.Surname = .TextBox1.Text
End With
If Ya.Name = "" Then
    i = MsgBox("Введите имя", vbOKCancel + vbCritical)
    If i = vbOK Then GoTo 10
End If

If Ya.Patronymic = "" Then
    i = MsgBox("Введите отчество", vbOKCancel + vbCritical)
    If i = vbOK Then GoTo 10
End If

If Ya.Surname = "" Then
    i = MsgBox("Введите фамилию", vbOKCancel + vbCritical)
    If i = vbOK Then GoTo 10
End If
MsgBox "Считано имя:" & vbCr & Ya.Name & " " & _
Ya.Patronymic & " " & Ya.Surname
End Sub
```

Вкладки

Шаг 1 | Шаг 2 | Шаг 3

Введите фамилию

Введите имя

Введите отчество

Пример, продолжение

Option Explicit

```
Private Sub CommandButton2_Click()  
Hide  
End Sub
```

```
Private Sub CommandButton3_Click()  
Hide  
End Sub
```

```
Private Sub CommandButton4_Click()  
Hide  
End Sub
```

```
Private Sub TabStrip1_Change()  
Dim i As Byte  
'Нужно сделать все формы невидимыми за исключением той  
'которая соответствует своей вкладке  
'Frame1 = Tab(0), Frame2=Tab(1), Frame2=Tab(2)  
With UserForm13  
For i = 1 To 3 'идём по Frame 1,2,3  
    .Controls.Item(i).Visible = False 'Frame - невидимы  
Next i  
' Делает видимым элемент Frmam(i) с индексом i, равным  
' номеру активной вкладки (value) с поправкой на 1  
    .Controls.Item(.TabStrip1.Value + 1).Visible = True  
End With  
End Sub
```

```
Private Sub UserForm_Activate()  
Frame1.Visible = True ' Пусть первая рамка будет видимой  
'TabStrip1.SelectedItem.Index = 0 'выделяем 1-ю закладку  
TabStrip1.Value = 0  
End Sub
```

```
Private Sub UserForm_Initialize()  
Dim i As Byte
```


```
UserForm13.Height = 110 'Задаём высоту формы  
With TabStrip1  
    .Left = 0 'координаты закладок  
    .Top = 0  
End With
```

```
For i = 1 To 3  
'Размещение рамок, которые на форме имеют индексы (TabIndex)  
'Frame1 - 1; Frame2-2; Frame3 - 3  
    With UserForm13.Controls.Item(i)  
        .Left = TabStrip1.ClientLeft 'расстояния с левого края  
        .Top = TabStrip1.ClientTop 'расстояние сверху  
        .Width = TabStrip1.ClientWidth 'ширина поля  
        .Height = TabStrip1.ClientHeight 'высота поля  
        .Visible = False  
    End With  
Next i
```

```
'Эквивалентная запись для каждой рамки:  
' Frame1.Left = TabStrip1.ClientLeft  
' Frame1.Top = TabStrip1.ClientTop  
' Frame1.Width = TabStrip1.ClientWidth  
' Frame1.Height = TabStrip1.ClientHeight  
  
End Sub
```

Дополнительные инструменты управления

В VBA, кроме стандартных элементов управления, имеется ряд дополнительных. Дополнительные элементы управления являются самостоятельными объектами, обладающие как общими для всех, так и индивидуальными свойствами и методами.



Календарь, свойства

Day	Возвращает выбранный день
DayFont, DayFontColor	Устанавливают шрифт и цвет шрифта для названий дней недели
DayLenght	Допустимые значения: <ul style="list-style-type: none">• Длинный (Long) (отображаются полные названия дней)• Средний (Medium) (в русской версии – отображаются две буквы из названий дней, в английской - три буквы)• Короткий (Short) (отображается только по первой букве из названия дня)
FirstDay	Первый день недели. Допустимые значения от воскресенье (Sunday) до суббота (Saturday)
Month	Возвращает выбранный месяц
MonthLenght	Допустимые значения: длинный (Long) (отображаются полные названия месяца) и короткий (short) (отображаются только первые три буквы из названия месяца)

Календарь, свойства и методы

ShowDays	Допустимые значения: True (отображаются названия дней недели) и False (в противном случае)
ShowDataselected	Допустимые значения: True (отображается выбранная дата в верхней части календаря) и False (в противном случае)
Value	Возвращает выбранную дату
Year	Возвращает выбранный год

Методы

NextDay, NextWeek, NextMonth и NextYear	Устанавливает следующий день, неделю, месяц и год
PreviousDay, PreviousWeek, PreviousMonth и PreviousYear	Устанавливает предыдущий день, неделю, месяц и год
Today	Устанавливает текущую дату в календаре

Передача элементов управления в подпрограммы

Элементы управления передаются как аргументы процедуры по ссылке. Тип задаётся с использованием класса **MSForms**

```
Private Sub tbxXC_Exit(ByVal Cancel As  
MSForms.ReturnBoolean)
```

```
VerifyNumber (tbxXC)
```

```
End Sub
```

```
Private Sub VerifyNumber(tbx As  
MSForms.TextBox)
```

```
If IsNumeric(tbx) Then Exit Sub
```

```
MsgBox "Error" & tbx, vbExclamation
```

```
tbx = ""
```

```
End Sub
```

СПАСИБО ЗА ВНИМАНИЕ!